



# An exercise on streams: convergence acceleration

Pierre Lescanne

► **To cite this version:**

| Pierre Lescanne. An exercise on streams: convergence acceleration. 2013. <ensl-00920026v2>

**HAL Id: ensl-00920026**

**<https://hal-ens-lyon.archives-ouvertes.fr/ensl-00920026v2>**

Submitted on 3 Mar 2014

**HAL** is a multi-disciplinary open access archive for the deposit and dissemination of scientific research documents, whether they are published or not. The documents may come from teaching and research institutions in France or abroad, or from public or private research centers.

L'archive ouverte pluridisciplinaire **HAL**, est destinée au dépôt et à la diffusion de documents scientifiques de niveau recherche, publiés ou non, émanant des établissements d'enseignement et de recherche français ou étrangers, des laboratoires publics ou privés.

# An exercise on streams: convergence acceleration

Pierre Lescanne  
University of Lyon,  
École normale supérieure de Lyon,  
LIP (UMR 5668 CNRS ENS Lyon UCBL INRIA)  
46 allée d'Italie, 69364 Lyon, France

## Abstract

---

This paper presents convergence acceleration, a method for computing efficiently the limit of numerical sequences as a typical application of streams and higher-order functions.

**Keywords:** convergence acceleration, streams, numerical analysis, co-algebra

---

## 1 Introduction

Assume that we want to compute numerically the limit of a sequence that converges slowly. If we use the sequence itself, we will get significant figures of the limit after a long time. Methods called *convergence acceleration* have been designed to shorten the time after which we get reasonable amount of significant digits. In other words, *convergence acceleration* is a set of methods for numerically computing the limit of a sequence of numbers. Those methods are based on sequence transformations and are a nice domain of application of streams [14, 1], with beautiful higher order functions. This allows us to present elegantly rather complex methods and to code them in Haskell [7] replacing long, obscure and special purpose Fortran programs by short, generic and arbitrary precision Haskell programs.

In this paper we show, how given a sequence  $(s_n)_{n \in \mathbb{N}}$ , we can evaluate efficiently  $\lim_{n \rightarrow \infty} s_n$ . For that we use *Levin transforms*. There are three kinds of such transforms, which are the result of three sequence transformations labeled traditionally by  $t$ ,  $u$  and  $v$  ([11], p.58).

## 2 Presentation of the method

In what follows we speak indistinctly of “sequences” or of “streams” We use Haskell and we work with arbitrary precision reals based on the implementation of David Lester called *CReal*. In Haskell the type of streams over a type  $A$  is written  $[A]$ .

For the numerical aspect, we follow Naoki Osada [11] (for a historical account see [4]) and we show that the stream notation of Haskell makes the presentation simpler. With no surprise, the size of the Haskell code is the same if not shorter than the mathematical description of the reference [11]. Moreover it provides efficient programs.

Levin transformations are somewhat generic in the sense that they are based on elementary transformations. Specialists of convergence acceleration propose three such elementary transformations. Let  $s$  be a sequence on *CReal*, i.e.,  $s :: [CReal]$ .

We define first a basic sequence transformation on which we will found our elementary transformations:

```
dELTA :: [CReal] → [CReal]
dELTA s = zipWith (-) (tail s) s
```

which means that  $dELTA(s)_n = s_{n+1} - s_n$ . From this basic sequence transformation we define the three elementary other sequence transformations as follows. A unique function depending on a parameter which is either  $T$  or  $U$  or  $V$  is given. For a computer scientist those names  $T$ ,  $U$ ,  $V$  are meaningless but this terminology (in lower case i.e.,  $u$ ,  $t$ ,  $v$  is traditionally used by mathematicians and we stick to it. It corresponds to the traditional notations of numerical analysis for those sequence transformations.

```
data Kind = T | U | V
```

```
delta :: Kind → [CReal] → [CReal]
delta T s = dELTA s
delta U s = zipWith (*) (dELTA s) [1..]
delta V s = zipWith (/) (zipWith (*) (tail $ dELTA s) (dELTA s))
              (dELTA (dELTA s))
```

In numerical analysis, people speak about *E-algorithm*. This is a family of functions  $eAlg_{n,k}$  which are also parametrized by a character either  $T$  or  $U$  or  $V$ . It tells which of the basic sequence transformations is chosen.  $eAlg_{n,k}$  uses a family of auxiliary functions which we call  $gAlg_{n,k}$  for symmetry and regularity. Here is the Haskell code for these functions:

```
eAlg :: Kind → Int → [CReal] → [CReal]
eAlg c 0 s = s
eAlg c k s = let
    a = (eAlg c (k - 1) s)
    b = (gAlg c (k - 1) k s)
  in
    zipWith (-) a (zipWith (*) b (zipWith (/) (dELTA a) (dELTA b)))

gAlg :: Kind → Int → Int → [CReal] → [CReal]
gAlg c 0 j s = let
    nTojMinus1 j = zipWith (**) [1..] (repeat (fromIntegral (j - 1)))
  in
    zipWith (/) (nTojMinus1 j) (delta c s)
gAlg c k j s = let
    a = gAlg c (k - 1) j s
    b = gAlg c (k - 1) k s
  in
    zipWith (-) a (zipWith (*) b (zipWith (/) (dELTA a) (dELTA b)))
```

Here is the formula as it is given in [11].  $R_n$  is the generic value of  $(delta\ c\ s)_n$ .  $gAlg$  is written  $g$ .  $E_k^{(n)}$  is the  $n^{\text{th}}$  element of the sequence  $eAlg\ c\ k\ s$ , the same for  $g_{k,j}^{(n)}$ .  $\Delta$  is the notation for what we write  $dELTA$ .

$$\begin{aligned}
E_0^{(n)} &= s_n, \quad g_{0,j}^{(n)} = n^{1-j} R_n, \quad n = 1, 2, \dots; \quad j = 1, 2, \dots, \\
E_k^{(n)} &= E_{k-1}^{(n)} - g_{k-1,k}^{(n)} \frac{\Delta E_{k-1}^{(n)}}{\Delta g_{k-1,k}^{(n)}}, \quad n = 1, 2, \dots; \quad k = 1, 2, \dots, \\
g_{k,j}^{(n)} &= g_{k-1,j}^{(n)} - g_{k-1,k}^{(n)} \frac{\Delta g_{k-1,j}^{(n)}}{\Delta g_{k-1,k}^{(n)}}, \quad n = 1, 2, \dots; \quad k = 1, 2, \dots, \quad j > k
\end{aligned}$$

### 3 Levin's formulas

There is another formula for the  $E$  – algorithm:

$$T_k^{(n)} = \frac{\sum_{j=0}^k (-1)^j \binom{k}{j} \binom{n+j}{n+k}^{k-1} \frac{s_{n+j}}{R_{n+j}}}{\sum_{j=0}^k (-1)^j \binom{k}{j} \binom{n+j}{n+k}^{k-1} \frac{1}{R_{n+j}}}$$

$T_k^{(n)}$  is not easily amenable to a Haskell program.<sup>1</sup> We give only the functions for  $k = 0, 1, 2$ , which we call *levin c 0*, *levin c 1* and *levin c 2*. For  $k = 0$ ,  $T_0^{(n)} = s_n$  and *levin k 1* and *levin k 2* are the result of small calculations.

```
levin :: Kind -> Int -> [CReal] -> [CReal]
levin c 0 s = s
levin c 1 s = zipWith (-) s (zipWith (/) (zipWith (*) (dELTA s) (delta c s))
                                         (dELTA (delta c s)))
```

*levin U 1 s* is called *Aitken's delta-squared process*. One notices that the numerator and the denominator of the above formula differ slightly. Indeed  $s_{n+j}$  in the numerator is just replaced by 1 in the denominator.

```
formulaForLevinTwo :: Kind -> [CReal] -> [CReal] -> [CReal]
formulaForLevinTwo c s' s =
  zipWith (+) (zipWith (-) (foldl (zipWith (*)) [2..] [tail$tail s', tail$delta c s, delta c s])
                          (foldl (zipWith (*)) [2, 4..] [tail s', tail$tail$delta c s, delta c s]))
             (foldl (zipWith (*)) [0..] [s', tail$tail$delta c s, tail$delta c s])
```

```
levin c 2 s = zipWith (/) (formulaForLevinTwo c s s) (forLevinTwo c [1, 1..] s)
```

Brezinski [3] proves that the sequences  $E_k^{(n)}$  and  $T_k^{(n)}$  are the same, in other words:

```
eAlg =η levin
```

### 4 Classic and non classic examples

*More than any other branch of numerical analysis, convergence acceleration is an experimental science. The researcher applies the algorithm and looks at the results to assess their worth.*

Dirk Laurie [8]

Given a sequence we use the convergence acceleration to find the main coefficient of the asymptotic equivalent. More precisely given an integer sequence  $s_n$ , we want to find a number  $a > 1$  such that  $s_n \sim a^n f(n)$  where  $f(n)$  is subexponential. In other words, for all  $b \in \mathbb{R}$ ,  $b > 1$ ,  $\lim_{n \rightarrow \infty} \frac{f(n)}{b^n} = 0$ . Actually we compute the limit of the sequence  $s_{n+1}/s_n$ . For that we create the function:

<sup>1</sup>In particular due to many divisions by 0 and to the complexity of the formula.

```

expCoeffAC :: ([CReal] → [CReal]) → [Integer] → Int → CReal
expCoeffAC transform sequence n = last (transform (zipWith (/) (tail u) u))
  where
    u = map fromIntegral (take n sequence)

```

Thus `expCoeffAC (levin U 2) s 300` gives the approximation of the coefficient one can get after 300 iterations using the sequence transformation `levin U 2`.

## 4.1 Catalan numbers

A good example to start is Catalan numbers:

```

catalan = 1 : [let cati = take i catalan
               in sum (zipWith (*) cati (reverse cati)) | i ← [1..]]

```

We know that

$$catalan!!n \sim \frac{4^n}{\sqrt{\pi n^3}}$$

Actually we get `expCoeffAC (levin U 2) catalan 800 = 4.0000000237` (with 8 exacts digits) and

$$\begin{aligned} \text{expCoeffAC (eAlg T 2) catalan 800} &\approx 3.9849561088 \\ \text{expCoeffAC (eAlg U 2) catalan 800} &\approx 3.9773868157 \\ \text{expCoeffAC (eAlg V 2) catalan 800} &\approx 3.9773869346 \end{aligned}$$

## 4.2 Counting plain lambda terms

Now we want to use this technique to address a conjecture<sup>2</sup> [6], on the asymptotic evaluation of the exponential coefficient of the numbers of typable terms of size  $n$  when  $n$  goes to  $\infty$ . First let us give the recursive definition of the numbers  $S_\infty$  of plain lambda terms of size  $n$ . This sequence appears on the *On-line Encyclopedia of Integer Sequences* with the entry number **A114851**. We assume that abstractions and applications have size 2 and variables have size  $1 + k$  where  $k$  is the depth of the variable w.r.t. its binder.

$$\begin{aligned} S_{\infty,0} &= S_{\infty,1} = 0, \\ S_{\infty,n+2} &= 1 + S_{\infty,n} + \sum_{k=0}^n S_{\infty,k} S_{\infty,n-k}. \end{aligned}$$

It has been proved in [6] that

$$S_{\infty,n} \sim A^n \cdot \frac{C}{n^{3/2}},$$

where  $A \doteq 1.963447954$  and  $C \doteq 1.021874073$ . After 300 iterations and using `levin U 2` we found

$$\mathbf{1.9634489522735283291619147713569993355616}.$$

giving six exact digits.

---

<sup>2</sup>This problem is the origin of the interest of the author for this question.

### 4.3 Counting typable lambda terms

The question is now to find the exponential coefficients for the numbers of typable terms. We have no formula for computing those numbers. The only fact we know is the following table of the numbers  $T_{\infty,n}$  till 42 which has been obtained after heavy computations (more than 5 days for the 42<sup>nd</sup>). The method consists in generating all the lambda terms of a given size and sieving those that are typable to count them. Therefore the best method to guess the exponential coefficient is by acceleration of convergence. After 43 iterations we found 1.8375065809.... Knowing that with the same number 43 we get 1.8925174623... for  $S_{\infty}$ , this is not enough to conclude. But this allows us to speculate that the exponential coefficient for the asymptotic evaluation of  $T_{\infty,n}$  could be 1.963447954 like for  $S_{\infty,n}$ 's.

## 5 Application to divergent series

In his famous paper [5] Euler provides a sum to divergent series. See [12, 9] and for a light introduction, the reader who understands French is advised to watch the video [13] which completes another video [10] in English.

Among the methods Euler and his followers propose to give a meaning to sum of divergent series there is convergence acceleration. We applied naturally our implementation to some divergent series.

Lets us define the function that associated to a sequence it series.

```
seq2series :: [Int] -> [CReal]
seq2series s = let series (x : s') = x : map ((+) x) (series s')
               in map fromIntegral (series s)
```

### 5.1 Grandi series

Grandi series is also called by Euler, Leibniz series. This is the series

$$\sum_{i=0}^{\infty} (-1)^i$$

which is sometime written  $1 - 1 + 1 - 1 + \dots$ . In Haskell it is:

```
grandi = let gr = 1 : (-1) : gr in seq2series gr
```

We get 1/2 after 3 iterations using *eAlg T 2* and this does not change when we increase the number of iterations.

### 5.2 Leibniz series 1 - 2 + 3 - 4 + 5 + ...

This series attributed to Leibniz is also studied by Euler:

$$\sum_{i=0}^{\infty} (-1)^{i+1} i$$

In Haskell

```
sumNatAlt = let lbn i = -i : (i + 1) : lbn (i + 2) in seq2series (lbn 0)
```

By using *eAlg U 4* we get 0.25 after 6 iterations in accordance with Euler's result  $\frac{1}{4}$ .

## 6 Conclusion

We have shown how streams can be applied to numerical analysis, namely to convergence acceleration. It makes no doubt that they can also be applied to other fields in numerical analysis or elsewhere. For instance, one may imagine applications of acceleration of convergence to the computation of limits of non numerical sequences.

We did not use the differential equation approach [14, 15], but presenting acceleration of convergence in this framework should be worthwhile.

## References

- [1] Marcello M. Bonsangue, Jan J. M. M. Rutten, and Joost Winter. Defining context-free power series coalgebraically. In Dirk Pattinson and Lutz Schröder, editors, *CMCS, volume 7399 of Lecture Notes in Computer Science*, pages 20–39. Springer, 2012.
- [2] Folkmar Bornemann, D. P. Dirk Pieter Laurie, S. Wagon, and Jörg Waldvogel. *The SIAM 100-digit challenge : a study in high-accuracy numerical computing*. Society for Industrial and Applied Mathematics, Philadelphia, 2004.
- [3] Claude Brezinski. A general extrapolation algorithm. *Numer. Math.*, 35, 1980.
- [4] Claude Brezinski. Convergence acceleration during the 20th century. *J. Comput. Appl. Math.*, 122:1–21, 2000.
- [5] L. Euler. De seriebus divergentibus. *Novi comm. acad. sci. Petrop.*, 5:205–237, 1760. in *Opera Omnia*, 14 (1) (1925), pp. 585-617 Leipzig and Berlin.
- [6] Katarzyna Grygiel and Pierre Lescanne. Counting terms in the binary lambda calculus. Technical report, ENS de Lyon, 2014. <http://hal-ens-lyon.archives-ouvertes.fr/ensl-00923203>.
- [7] Graham Hutton. *Programming in Haskell*. Cambridge University Press, 2007.
- [8] D. P. Dirk Pieter Laurie. *Appendix A: Convergence acceleration*, pages 227–261. In [2], 2004.
- [9] J. Martinet and J.-P. RAMIS. Elementary acceleration and multisummability. *Ann. Inst. Henri Poincaré, Physique Théorique*, 54(1), 1991.
- [10] Numberphile. Astounding:  $1 + 2 + 3 + 4 + 5 + \dots = -1/12$ . <http://www.youtube.com/watch?v=w-I6XTVZXww>, January 2014. video.
- [11] Naoki Osada. *Acceleration Methods for Slowly Convergent Sequences and their Applications*. PhD thesis, Nagoya University, 1993.
- [12] J.-P. RAMIS. *Séries divergentes et théories asymptotiques*. Panoramas et Synthèses. Société mathématique de France, 1993.
- [13] Jean-Pierre Ramis. Leonhard euler, ou l’art de donner un sens ce qui n’en avait pas. <http://vimeo.com/30907930>, April 2009. video.
- [14] Jan J. M. M. Rutten. Behavioural differential equations: a coinductive calculus of streams, automata, and power series. *Theor. Comput. Sci.*, 308(1-3):1–53, 2003.
- [15] Joost Winter. QStream: A suite of streams. In Reiko Heckel and Stefan Milius, editors, *CALCO*, volume 8089 of *Lecture Notes in Computer Science*, pages 353–358. Springer, 2013.